{-# OPTIONS_GHC -O2 #-}
{-# LANGUAGE BangPatterns, BinaryLiterals, CPP, DerivingStrategies  #-}
{-# LANGUAGE DerivingVia, FlexibleContexts, FlexibleInstances       #-}
{-# LANGUAGE GeneralizedNewtypeDeriving, KindSignatures, LambdaCase #-}
{-# LANGUAGE MagicHash, MultiParamTypeClasses, MultiWayIf           #-}
{-# LANGUAGE NumericUnderscores, OverloadedStrings, PatternSynonyms #-}
{-# LANGUAGE RankNTypes, RecordWildCards, ScopedTypeVariables       #-}
{-# LANGUAGE StandaloneDeriving, TupleSections, TypeApplications    #-}
{-# LANGUAGE TypeFamilies, TypeInType, UnboxedTuples, ViewPatterns  #-}

import           Control.Applicative
import           Control.Exception
import           Control.Monad
import           Control.Monad.Primitive
import           Control.Monad.Reader
import           Control.Monad.ST
import           Control.Monad.State.Strict
import           Data.Bifunctor
import           Data.Bool
import qualified Data.ByteString                   as B
import qualified Data.ByteString.Builder           as B
import qualified Data.ByteString.Char8             as C
import qualified Data.ByteString.Internal          as B
import qualified Data.ByteString.Unsafe            as B
import           Data.Char
import qualified Data.Foldable                     as F
import           Data.Function
import           Data.Functor.Identity
import qualified Data.IntMap.Strict                as IM
import qualified Data.IntSet                       as IS
import qualified Data.List                         as L
import qualified Data.Map.Strict                   as M
import           Data.Monoid
import           Data.Ord
import           Data.Primitive
import           Data.Proxy
import           Data.Ratio
import           Data.Semigroup
import qualified Data.Set                          as S
import           Data.Tuple
import qualified Data.Vector                       as V
import qualified Data.Vector.Algorithms.Intro      as Intro
import qualified Data.Vector.Fusion.Stream.Monadic as MS
import qualified Data.Vector.Fusion.Bundle.Monadic as MB
import           Data.Vector.Fusion.Util
import qualified Data.Vector.Generic               as G
import qualified Data.Vector.Generic.Mutable       as GM
import qualified Data.Vector.Mutable               as VM
import qualified Data.Vector.Primitive             as P
import qualified Data.Vector.Primitive.Mutable     as PM
import qualified Data.Vector.Unboxed               as U
import qualified Data.Vector.Unboxed.Mutable       as UM
import           Debug.Trace
import           Foreign                           hiding (void)
import           GHC.Exts
import           GHC.TypeLits
import           System.IO
import           Unsafe.Coerce

main :: IO ()
main = do
    [n, m] <- map read.words <$> getLine
    xs <- U.unfoldrN n (runParser int) <$> C.getLine
    ys <- U.unfoldrN m (runParser int) <$> C.getLine
    putStr.unwords.map show.U.toList $ solve n m xs ys

solve :: Int -> Int -> U.Vector Int -> U.Vector Int -> U.Vector Int
solve n m xs ys = convolute 998244353 3 xs ys

data NTTRunner = NTTRunner
    { pNR :: !Int
    , gNR :: !Int
    , ipNR :: !Word
    , sesNR :: !(U.Vector Int)
    , siesNR :: !(U.Vector Int)
    }

buildNTTRunner :: Int -> Int -> NTTRunner
buildNTTRunner pNR gNR = NTTRunner{..}
  where
    ipNR = quot (complement 0) (fromIntegral pNR) + 1

    ctz = countTrailingZeros (pNR - 1)
    !e = powMod gNR (unsafeShiftR (pNR - 1) ctz) pNR
    !ie = recipMod e pNR

    es  = U.reverse $ U.iterateN (ctz - 1) (\x -> x *% x) e
    ies = U.reverse $ U.iterateN (ctz - 1) (\x -> x *% x) ie

    sesNR = U.zipWith (*%) es $ U.scanl' (*%) 1 ies
    siesNR = U.zipWith (*%) ies $ U.scanl' (*%) 1 es
    x *% y = x * y `rem` pNR

addNR :: NTTRunner -> Int -> Int -> Int
addNR (NTTRunner (I# m#) _ _ _ _) (I# x#) (I# y#)
    = case x# +# y# of
        z# -> I# (z# -# ((z# >=# m#) *# m#))
{-# INLINE addNR #-}

subNR :: NTTRunner -> Int -> Int -> Int
subNR (NTTRunner (I# m#) _ _ _ _) (I# x#) (I# y#)
    = case x# -# y# of
        z# -> I# (z# +# ((z# <# 0#) *# m#))
{-# INLINE subNR #-}

mulNR :: NTTRunner -> Int -> Int -> Int
mulNR (NTTRunner (I# m0#) _ (W# im#) _ _) (I# x#) (I# y#)
    = case timesWord# (int2Word# x#) (int2Word# y#) of
        z# -> case timesWord2# z# im# of
            (# q#, _ #) -> case minusWord# z# (timesWord# q# m#) of
                v# | isTrue# (geWord# v# m#) -> I# (word2Int# (plusWord# v# m#))
                   | otherwise -> I# (word2Int# v#)
  where
    m# = int2Word# m0#
{-# INLINE mulNR #-}

butterfly :: (PrimMonad m) => NTTRunner -> UM.MVector (PrimState m) Int -> m ()
butterfly nr@NTTRunner{..} mvec = do
    flip MS.mapM_ (stream 1 (h + 1)) $ \ph -> do
        let !w = unsafeShiftL 1 (ph - 1)
            !p = unsafeShiftL 1 (h - ph)
        void $ MS.foldlM'
            (\acc s -> do
                let offset = unsafeShiftL s (h - ph + 1)
                flip MS.mapM_ (stream offset (offset + p)) $ \i -> do
                    l <- UM.unsafeRead mvec i
                    r <- mulNR nr acc <$!> UM.unsafeRead mvec (i + p)
                    UM.unsafeWrite mvec (i + p) $ subNR nr l r
                    UM.unsafeWrite mvec i $ addNR nr l r
                return $! mulNR nr acc $ U.unsafeIndex siesNR (countTrailingZeros (complement s))
            ) 1 (stream 0 w)
  where
    n = UM.length mvec
    !h = head [i | i<-[0..], n <= unsafeShiftL 1 i]
{-# INLINE butterfly #-}

invButterfly :: (PrimMonad m) => NTTRunner -> UM.MVector (PrimState m) Int -> m ()
invButterfly nr@NTTRunner{..} mvec = void $ do
    flip MS.mapM_ (streamR 1 (h + 1)) $ \ph -> do
        let !w = unsafeShiftL 1 (ph - 1)
            !p = unsafeShiftL 1 (h - ph)
        MS.foldlM'
            (\acc s -> do
                let offset = unsafeShiftL s (h - ph + 1)
                flip MS.mapM_ (stream offset (offset + p)) $ \i -> do
                    l <- UM.unsafeRead mvec i
                    r <- UM.unsafeRead mvec (i + p)
                    UM.unsafeWrite mvec (i + p) $ mulNR nr acc (pNR + l - r)
                    UM.unsafeWrite mvec i $ addNR nr l r
                return $! mulNR nr acc $ U.unsafeIndex sesNR (countTrailingZeros (complement s))
            ) 1 (stream 0 w)
  where
    n = UM.length mvec
    !h = head [i | i<-[0..], n <= unsafeShiftL 1 i]
{-# INLINE invButterfly #-}

convolute :: Int -> Int -> U.Vector Int -> U.Vector Int -> U.Vector Int
convolute p g xs ys = U.create $ do
    mxs <- UM.replicate len 0
    U.unsafeCopy (UM.take n mxs) xs
    butterfly nr mxs
    mys <- UM.replicate len 0
    U.unsafeCopy (UM.take m mys) ys
    butterfly nr mys
    rep len $ \i -> do
        yi <- UM.unsafeRead mys i
        UM.unsafeModify mxs (mulNR nr yi) i
    invButterfly nr mxs
    rep (n + m - 1) $ \i -> do
        UM.unsafeModify mxs (mulNR nr ilen) i
    return $ UM.take (n + m - 1) mxs
  where
    !nr = buildNTTRunner p g
    n = U.length xs
    m = U.length ys
    !h = head [i | i<-[0..], n + m - 1 <= unsafeShiftL 1 i]
    !len = unsafeShiftL 1 h
    !ilen = recipMod len p
{-# INLINE convolute #-}

powMod :: (Integral a) => a -> Int -> a -> a
powMod x n m
    | n > 0 = go 1 x n
    | n == 0 = 1
    | otherwise = go 1 (recipMod x m) (-n)
  where
    go !acc !y !i
        | i .&. 1 == 0 = go acc (y * y `rem` m) (unsafeShiftR i 1)
        | i == 1 = acc * y `mod` m
        | otherwise = go (acc * y `rem` m) (y * y `rem` m) (unsafeShiftR (i - 1) 1)
{-# INLINE powMod #-}

recipMod :: (Integral a) => a -> a -> a
recipMod x m = go x m 1 0
  where
    go !a !b !u !v
        | b > 0 = case a `quot` b of
            q -> go b (a - (q * b)) v (u - (q * v))
        | otherwise = u `mod` m
{-# INLINE recipMod #-}


-------------------------------------------------------------------------------
-- Utils
-------------------------------------------------------------------------------
rep :: (Monad m) => Int -> (Int -> m ()) -> m ()
rep n = flip MS.mapM_ (stream 0 n)
{-# INLINE rep #-}
rev :: (Monad m) => Int -> (Int -> m ()) -> m ()
rev !n = flip MS.mapM_ (streamR 0 n)
{-# INLINE rev #-}
stream :: (Monad m) => Int -> Int -> MS.Stream m Int
stream !l !r = MS.Stream step l where { step x | x < r = return $ MS.Yield x (x + 1) | otherwise = return MS.Done; {-# INLINE [0] step #-}}
{-# INLINE [1] stream #-}
streamR :: (Monad m) => Int -> Int -> MS.Stream m Int
streamR !l !r = MS.Stream step (r - 1) where { step x | x >= l = return $ MS.Yield x (x - 1) | otherwise = return MS.Done; {-# INLINE [0] step #-}}
{-# INLINE [1] streamR #-}
stream' :: (Monad m) => Int -> Int -> Int -> MS.Stream m Int
stream' !l !r !d = MS.Stream step l where { step x | x < r = return $ MS.Yield x (x + d) | otherwise = return MS.Done; {-# INLINE [0] step #-}}
{-# INLINE [1] stream' #-}
infixl 8 `shiftRL`, `unsafeShiftRL`
shiftRL :: Int -> Int -> Int
shiftRL = unsafeShiftRL
{-# INLINE shiftRL #-}
unsafeShiftRL :: Int -> Int -> Int
unsafeShiftRL (I# x#) (I# i#) = I# (uncheckedIShiftRL# x# i#)
{-# INLINE unsafeShiftRL #-}
type Parser a = StateT C.ByteString Maybe a
runParser :: Parser a -> C.ByteString -> Maybe (a, C.ByteString)
runParser = runStateT
{-# INLINE runParser #-}
int :: Parser Int
int = coerce $ C.readInt . C.dropWhile isSpace
{-# INLINE int #-}
int1 :: Parser Int
int1 = fmap (subtract 1) int
{-# INLINE int1 #-}
char :: Parser Char
char = coerce C.uncons
{-# INLINE char #-}
byte :: Parser Word8
byte = coerce B.uncons
{-# INLINE byte #-}
skipSpaces :: Parser ()
skipSpaces = modify' (C.dropWhile isSpace)
{-# INLINE skipSpaces #-}
lowerBoundM :: (Monad m) => Int -> Int -> (Int -> m Bool) -> m Int
lowerBoundM low high p = go low high where { go !low !high | high <= low = return high | otherwise = p mid >>= bool (go (mid + 1) high) (go low mid) where { mid = low + unsafeShiftRL (high - low) 1}}
{-# INLINE lowerBoundM #-}
upperBoundM :: (Monad m) => Int -> Int -> (Int -> m Bool) -> m Int
upperBoundM low high p = do { flg <- p high; if flg then return high else subtract 1 <$!> lowerBoundM low high (fmap not . p)}
{-# INLINE upperBoundM #-}
lowerBound :: Int -> Int -> (Int -> Bool) -> Int
lowerBound low high p = runIdentity (lowerBoundM low high (return . p))
{-# INLINE lowerBound #-}
upperBound :: Int -> Int -> (Int -> Bool) -> Int
upperBound low high p = runIdentity (upperBoundM low high (return . p))
{-# INLINE upperBound #-}


Measured memory_bytes: 57220